

Vorlesung Architekturen und Entwurf von Rechnersystemen



TECHNISCHE
UNIVERSITÄT
DARMSTADT

Lösungsvorschlag

Prof. Andreas Koch, Yannick Lavan, Johannes Wirth, Mihaela Damian

Wintersemester 2022/2023
Übungsblatt 2

In dieser Übung lernen Sie folgende Konzepte und ihre Umsetzung in BSV kennen:

- Subinterfaces
- FIFOs
- FSMs
- Tagged Unions

Aufgabe 2.1: Einfache Pipeline

Wir wollen eine simple Berechnungspipeline implementieren, welche die Formel $res = (((x + a) \cdot b) \cdot c) / 4 + 128$ berechnet. Dabei ist x ein Eingabedatum und a, b, c sind Konfigurationsparameter, welche wir zur Laufzeit im Modul ändern wollen. Wir nehmen hierfür an, dass alle Operationen in einem Takt ausgeführt werden.




Abbildung 1: Pipeline aus Aufgabe 2.1.

Aus objektorientierten Programmiersprachen kennen Sie die Möglichkeit, dass eine Klasse mehrere Interfaces implementiert. Dieses Konzept macht man sich auch im Hardware-Entwurf zunutze und verwendet dafür in BSV das Konzept der Subinterfaces. Jedes Interface, das in Bluespec deklariert wurde, kann als Teil eines größeren Interfaces verwendet werden. Dies ermöglicht es, ein Modul mit mehreren anderen Modulen kommunizieren zu lassen oder ein Modul mit zusätzlicher Funktionalität zu erweitern, es aber in vereinfachter Form zu behandeln. Letztere Variante betrachten wir in dieser Aufgabe.

Wenn wir die Pipeline in Abbildung 1 betrachten sehen wir, dass das zu implementierende Modul folgende Funktionalitäten bereitstellen muss:

1. Input des Wertes x
2. Output des Ergebnisses
3. Setzen der Parameter a, b, c .

Hierbei kann Unterschieden werden in Funktionalität zum Ein- und Ausgeben der Daten (Punkte 1 & 2) und Funktionalität zum Konfigurieren der Parameter (Punkt 3). Für eine modulare Gestaltung wollen wir die Punkte 1 & 2 in ein eigenes Interface auslagern. Dieses werden wir dann später an anderer Stelle wiederverwenden können. Dieses Interface können wir wie folgt in BSV formalisieren:

```
interface CalcUnit;
  method Action put(Int#(32) x);
  method ActionValue#(Int#(32)) result();
endinterface
```

Das Interface für unsere Pipeline sieht dann wie folgt aus:

```
interface Pipeline;
  interface CalcUnit calc;
  method Action setParam(UInt#(2) addr, Int#(32) val);
endinterface
```

Dieses Interface verwendet das vorher definierte Interface **CalcUnit** als Subinterface und erlaubt das Ein- und Ausgeben der Daten. Zusätzlich ermöglicht es das Konfigurieren der Parameter über die Methode **setParam()**. Dabei soll für **addr** = 0 der Wert von *a* gesetzt werden, für **addr** = 1 der Wert von *b* und für **addr** = 2 der Wert von *c*.

Implementieren Sie nun ein entsprechendes Modul **mkSimplePipeline**, welches das Interface **Pipeline** implementiert und die oben beschriebene Formel berechnet. Nutzen Sie dabei zum Speichern der Eingaben und Ausgaben FIFOs, zum Zwischenspeichern der Ergebnisse Register (wie in Abbildung 1 dargestellt). Sie können sich dabei an der Implementierung der Shifter Pipeline aus der Vorlesung orientieren (2. Realisierung)¹. Welchen Typ hat die soeben implementierte Pipeline (dynamisch vs statisch; elastisch vs starr)?

Lösungsvorschlag:

Die implementierte Pipeline ist **statisch** und **starr**.

```
module mkPipeline(Pipeline);
  FIFO#(Int#(32)) fifo_in <- mkFIFO;
  FIFO#(Int#(32)) fifo_out <- mkFIFO;
  Vector#(3, Reg#(Int#(32))) params <- replicateM(mkRegU);
  Vector#(4, Reg#(Maybe#(Int#(32)))) regs <- replicateM(mkReg(tagged Invalid));

  rule all_together;
    // Stage 0
    if (fifo_in.notEmpty) begin
      let x = fifo_in.first; fifo_in.deq;
      regs[0] <= tagged Valid (x + params[0]);
    end else begin
      regs[0] <= tagged Invalid;
    end

    // Stage 1
    if (regs[0] matches tagged Valid .x) begin
      regs[1] <= tagged Valid (x * params[1]);
    end else begin
      regs[1] <= tagged Invalid;
    end

    // Stage 2
    if (regs[1] matches tagged Valid .x) begin
      regs[2] <= tagged Valid (x * params[2]);
    end
```

¹Foliensatz BSV; Folien 99-103

```

end else begin
    regs[2] <= tagged Invalid;
end

// Stage 3
if (regs[2] matches tagged Valid .x) begin
    regs[3] <= tagged Valid (x / 4); // alternatively >> 2
end else begin
    regs[3] <= tagged Invalid;
end

// Stage 4
if (regs[3] matches tagged Valid .x) begin
    fifo_out.enq(x + 128);
end
endrule

method Action setParam(UInt#(2) addr, Int#(32) val);
    params[addr] <= val;
endmethod

interface CalcUnit calc;
    method Action put(Int#(32) x);
        fifo_in.enq(x);
    endmethod

    method ActionValue#(Int#(32)) result();
        fifo_out.deq();
        return fifo_out.first;
    endmethod
endinterface
endmodule

```

Aufgabe 2.2: Testen mit FSMs

In der vorherigen Übung haben wir eine einfache regelbasierte Testbench verwendet. Wenn unsere Testbench aber viele kurze Schritte sequenziell abarbeiten soll, wird dieses Regelkonstrukt schnell unübersichtlich. Eine elegantere Alternative stellt die Nutzung des Packages **StmtFSM**² dar. Dabei erfolgt die Erstellung einer FSM in zwei Schritten:

1. Erstellung eines Statements, das den zeitlichen Ablauf der FSM beschreibt
2. Instanziierung einer FSM mit dem Statement aus (1)

2.2a) Erstellung des Statements

Machen Sie sich zunächst mit dem **StmtFSM**-Package vertraut. Sie finden einige Beispiele zur Nutzung in *BSV by Example*² und die Grammatik zur Konstruktion von Statements im *BSV Reference Guide*^{3,4}

In der Datei **U2_2_template.bsv** finden Sie das Codegerüst der Testbench für unser Pipeline-Modul. Wir geben Ihnen einen Vector mit gültigen Testergebnissen vor. Erstellen Sie in der Testbench ein Statement, das folgendes Verhalten abbildet:

1. Der Parameter *a* soll auf den Wert 42 gesetzt werden

²BSV by Example; Kapitel 14.3; S. 185ff

³BSV Language Reference Guide; Kapitel 12; S. 99ff.

⁴BSV Library Reference Guide; Kapitel 3.6; S. 172ff.

2. Der Parameter b soll auf den Wert 2 gesetzt werden
3. Der Parameter c soll auf den Wert 13 gesetzt werden
4. Anschließend soll eine Nachricht mit dem Inhalt „Initialized parameters, starting test...“ ausgegeben werden.
5. Nun sollen die folgenden zwei Auflistungen parallel ablaufen:
 - a)
 - Die Pipeline wird zehn Takte lang mit Testeingaben gefüttert. Diese sind die Zahlen 0 bis einschließlich 9.
 - Anschließend wird a auf den Wert 7 gesetzt, b auf den Wert 3 und c auf den Wert 17.
 - Daraufhin soll die Pipeline wieder mit Testeingaben gefüttert werden, den Zahlen 10 bis einschließlich 19.
 - b)
 - Die FSM soll 20 mal:
 - Die Nachricht „Checking results for INPUT“ ausgeben, wobei INPUT mit dem tatsächlichen Input-Wert ersetzt wird.
 - Das entsprechende Ergebnis aus der Pipeline auslesen
 - Überprüfen ob das ausgelesene Ergebnis mit dem passenden Eintrag aus **testvec** übereinstimmt und entsprechend einen Zähler für korrekte Tests um 1 erhöhen.
 - Sollte das ausgelesene Ergebnis Falsch sein soll die Nachricht „Expected CORRECT, got ACTUAL“ ausgegeben werden, wobei CORRECT mit dem entsprechenden Eintrag aus **testvec** ersetzt wird und ACTUAL mit dem zuvor ausgelesenen Ergebnis.
 - Dies soll alles in einem Takt geschehen.
6. Anschließend soll die Nachricht „N_CORR of 20 tests passed.“ ausgegeben werden, wobei N_CORR mit der Anzahl der korrekten Tests ersetzt werden soll.
7. Wenn alle Tests erfolgreich waren, soll anschließend die Nachricht „ SUCCESS!“ ausgegeben werden, ansonsten die Nachricht „FAILURE“.
8. Zuletzt soll die Simulation mit **\$finish()** beendet werden.

Hinweis: Machen Sie sich anhand von <http://wiki.bluespec.com/Home/Data-Types/Data-Type-Conversion-Functions> klar, wie Sie einen UInt zu einem Int konvertieren können.

Lösungsvorschlag:

```
Stmt s = seq
  dut.setParam(0, 42);
  dut.setParam(1, 2);
  dut.setParam(2, 13);
  $display("Initialized parameters, starting test...");
  par
    seq
      for(idx_put <= 0; idx_put < 10; idx_put <= idx_put + 1) action
        dut.calc.put(unpack(pack(idx_put)));
        // unpack(pack()) to cast from uint to int
      endaction
      dut.setParam(0, 7);
      dut.setParam(1, 3);
      dut.setParam(2, 17);
      for(idx_put <= 10; idx_put < 20; idx_put <= idx_put + 1) action
        dut.calc.put(unpack(pack(idx_put)));
      endaction
    endseq
```

```

seq
  for(idx_get <= 0; idx_get < 20; idx_get <= idx_get + 1) action
    $display("Checking result for %d", idx_get);
    let t <- dut.calc.result();
    if(t == testvec[idx_get])
      correct_tests <= correct_tests + 1;
    else $display("Expected %d, got %d.", testvec[idx_get], t);
  endaction
endseq
endpar
$display("%d of 20 tests passed.", correct_tests);
if(correct_tests == 20) $display("SUCCESS!");
else $display("FAILURE!");
$finish();
endseq;

```

2.2b) Instanziierung der FSM

Wir können **StmtFSMs** auf verschiedene Arten instanziiieren.

1. FSM fsm <- mkFSM(stmt) - Diese FSM können wir beliebig oft mit **fsm.start()** innerhalb einer Rule starten und mit **fsm.done()** prüfen, ob sie ihr Statement abgearbeitet hat.
2. mkAutoFSM(stmt) - Diese FSM wird direkt bei der Instanziierung gestartet und kann nicht nochmal ausgeführt werden.
3. FSM fsm <- mkFSMWithPred(stmt, bool) - Diese FSM verhält sich wie mkFSM, mit dem Unterschied, dass sie nach dem Start nur ausgeführt wird wenn die Bedingung **bool** erfüllt ist.

Instanziiieren Sie Ihre FSM mittels **mkAutoFSM**. Sie können Ihre Testbench nun kompilieren und Ihre Pipeline aus Aufgabe 2.1 testen. Sollte der Simulationsprozess irgendwo stecken bleiben, feuern Teile Ihrer Regeln oder Methoden nicht. Sie können sich die CAN_FIRE-Bedingungen Ihrer Regeln und Guards Ihrer Methoden anzeigen lassen, indem Sie dem ersten Compiler-Aufruf das Flag **-show-schedule** übergeben. Die Schedule Datei hat die Endung **.sched** und die CAN_FIRE steht hinter dem Punkt **Predicate**.

Hinweis: Damit ihre implementierte Pipeline korrekt funktioniert, müssen sie **bsc** möglicherweise mit dem Flag **-aggressive-conditions** aufrufen.

Lösungsvorschlag:

```
mkAutoFSM(s);
```

2.2c) Auswertung von FSMs

Gegeben sei das nachfolgende Modul mit drei verschiedenen FSM-Statements. Werten Sie die Statements *auf dem Papier* aus und geben Sie für jedes Statement die angezeigten Werte des Registers **ctr** an oder, ob es in dem Statement zu einem Compilerfehler kommt (und warum). Betrachten Sie die Statements individuell, d.h. Veränderungen durch die anderen Statements dürfen ignoriert werden.

```

1 module mkFSMEx(Empty);
2   Reg#(int) ctr <- mkReg(0);
3
4   Stmt s1 = seq
5     $display("%t : %d", $time(), ctr);
6     ctr <= ctr + 1;
7     $display("%t : %d", $time(), ctr);
8   par
9     seq
10      $display("%t : %d", $time(), ctr);
11    endseq
12    seq
13      action
14        ctr <= ctr + 1;
15        $display("%t : %d", $time(), ctr);
16      endaction
17      $display("%t, %d", $time(), ctr);
18    endseq
19  endpar
20 endseq;
21
22 Stmt s2 = seq
23   $display("%t : %d", $time(), ctr);
24   ctr <= ctr + 1;
25   $display("%t : %d", $time(), ctr);
26 par
27   seq
28     $display("%t : %d", $time(), ctr);
29   endseq
30   seq
31     ctr <= ctr + 1;
32     $display("%t : %d", $time(), ctr);
33     $display("%t, %d", $time(), ctr);
34   endseq
35 endpar
36 endseq;
37
38 Stmt s3 = seq
39   $display("%t : %d", $time(), ctr);
40   ctr <= ctr + 1;
41   $display("%t : %d", $time(), ctr);
42 par
43   seq
44     $display("%t : %d", $time(), ctr);
45   action
46     ctr <= ctr + 1;
47     ctr <= ctr + 1;
48   endaction
49   $display("%t : %d", $time(), ctr);
50 endseq
51 seq
52   ctr <= ctr + 1;
53   $display("%t : %d", $time(), ctr);
54   $display("%t, %d", $time(), ctr);
55 endseq
56 endpar
57 endseq;
58 endmodule

```

Lösungsvorschlag:

s1:

- Z. 5: 0
- Z. 7: 1
- Z. 10: 1
- Z. 15: 1
- Z. 17: 2

s2:

- Z. 23: 0
- Z. 25: 1
- Z. 28: 1
- Z. 32: 2
- Z. 33: 2

s3:

Für **s3** würde der Compiler einen Fehler werfen, da wir in Z. 51 und Z. 52 versuchen das **ctr**-Register im gleichen Takt zweimal zu schreiben.

Aufgabe 2.3: Modulare Pipeline

Wir wollen nun eine alternative Implementierung für die Pipeline aus Aufgabe 2.1 erstellen. Dabei wollen wir insbesondere die Berechnungen der einzelnen Pipelineinstufen in separate (Unter-)Module ("Calc Units") auslagern.

2.3a) Calc Units

Zuerst implementieren wir die für die einzelnen Berechnungen zuständigen Module.

Dabei gibt es zwei Typen von Berechnungen: Die letzten beiden Berechnungen (vgl. Abbildung 1) benötigen **keinen** Parameter. Daher können wir hier das bereits bekannte Interface **CalcUnit** nutzen:

```
interface CalcUnit;
  method Action put(Int#(32) x);
  method ActionValue#(Int#(32)) result();
endinterface
```

Die anderen drei Berechnungen benötigen jeweils einen Parameter (a , b bzw. c). Wir benötigen also ein neues Interface:

```
interface CalcUnitChangeable;
  interface CalcUnit calc;
  method Action setParameter(Int#(32) param);
endinterface
```

Beachten Sie, dass wir hier das Interface **CalcUnit** wiederverwenden, anstatt dessen zwei Methoden auch in das neue Interface zu schreiben.

In **U2_3_template.bsv** finden Sie die oben deklarierten Interfaces. Benennen Sie die Datei zunächst korrekt um. Implementieren Sie anschließend die Module in Tabelle 1 mit den passenden Interfaces. Dabei sollen die konfigurierbaren Parameter a , b und c jeweils in einem Register gespeichert werden. Sie können Wires⁵⁶ verwenden, um die Berechnungen mit kombinatorischer Logik umzusetzen. Sollten Sie sich für die kombinatorische Implementierung entscheiden, müssen Sie die Action **noAction**; in der Definition der Methode **result** aufrufen, um eine gültige ActionValue-Methode zu implementieren. Warum müssen Sie nur ein Multiplikationsmodul implementieren?

Modul	Interface
mkAddA	CalcUnitChangeable
mkMul	CalcUnitChangeable
mkDiv4	CalcUnit
mkAdd128	CalcUnit

Tabelle 1: Teilmodule und zugehörige Interfaces.

Lösungsvorschlag:

Es ist nur ein Multiplikationsmodul notwendig, da dieses mit unterschiedlichen Parametern wiederverwendet werden kann.

```
module mkAddA(CalcUnitChangeable);
  Reg#(Int#(32)) a <- mkRegU;
  Wire#(Int#(32)) in <- mkWire;

  method Action setParameter(Int#(32) param);
    a <= param;
```

⁵BSV by Example; Kapitel 8.2; S. 114ff.

⁶BSV Libraries Reference Guide; Kapitel 2.4.4; S. 51ff.

```

endmethod

interface CalcUnit calc;
  method Action put(Int#(32) x);
    in <= x;
  endmethod

  method ActionValue#(Int#(32)) result();
    noAction;
    return in + a;
  endmethod
endinterface
endmodule

module mkMul(CalcUnitChangeable);
  Reg#(Int#(32)) b <- mkRegU; // or c
  Wire#(Int#(32)) in <- mkWire;

  method Action setParameter(Int#(32) param);
    b <= param;
  endmethod

  interface CalcUnit calc;
    method Action put(Int#(32) x);
      in <= x;
    endmethod

    method ActionValue#(Int#(32)) result();
      noAction;
      return in * b;
    endmethod
  endinterface
endmodule

module mkDiv4(CalcUnit);
  Wire#(Int#(32)) in <- mkWire;

  method Action put(Int#(32) x);
    in <= x;
  endmethod

  method ActionValue#(Int#(32)) result();
    noAction;
    return in / 4; // alternatively >> 2
  endmethod
endmodule

module mkAdd128(CalcUnit);
  Wire#(Int#(32)) in <- mkWire;

  method Action put(Int#(32) x);
    in <= x;
  endmethod

  method ActionValue#(Int#(32)) result();

```



```

        noAction;
        return in + 128;
    endmethod
endmodule

```

2.3b) Pipeline Modul

In dieser Teilaufgabe wollen wir nun die einzelnen Teilmodule zu einer modularen Pipeline kombinieren. Diese soll das **Pipeline**-Interface aus Aufgabe 2.1 implementieren:

```

interface Pipeline;
    interface CalcUnit calc;
    method Action setParam(UInt#(2) addr, Int#(32) val);
endinterface

```

Implementieren Sie das entsprechende Modul.

Hinweis: Sie können die Pipeline-Funktionalität jetzt nicht mehr in einer einzelnen Regel implementieren, sondern müssen die Funktionalität auf mehrere Regeln aufteilen. Sie können Ihren Code mit geschickter Nutzung von for-Schleifen⁷⁸ und **Vector**⁹ deutlich kompakter halten.

Testen Sie die modulare Pipeline mit der Testbench aus Aufgabe 2.2. Sie sollten dafür keine Änderungen an der Testbench vornehmen müssen - abgesehen vom Tausch des DUT.

Lösungsvorschlag:

```

module mkPipeline(Pipeline);
    CalcUnitChangeable stage0 <- mkAddA;
    CalcUnitChangeable stage1 <- mkMul;
    CalcUnitChangeable stage2 <- mkMul;
    CalcUnit stage3 <- mkDiv4;
    CalcUnit stage4 <- mkAdd128;

    Vector#(3, CalcUnitChangeable) changeables;
    changeables[0] = stage0;
    changeables[1] = stage1;
    changeables[2] = stage2;

    Vector#(5, CalcUnit) stages;
    stages[0] = stage0.calc;
    stages[1] = stage1.calc;
    stages[2] = stage2.calc;
    stages[3] = stage3;
    stages[4] = stage4;

    FIFO#(Int#(32)) fifo_in <- mkFIFO;
    FIFO#(Int#(32)) fifo_out <- mkFIFO;
    Vector#(4, Array#(Reg#(Maybe#(Int#(32))))) regs <- replicateM(mkCReg(2, tagged
→ Invalid));

    function Bool setParamAllowed();

```

⁷BSV by Example; Kapitel 11; S. 151ff

⁸BSV Language Reference Guide; Kapitel 9.7; S. 74f.

⁹BSV by Example; Kapitel 13; S. 163ff.

```

    Bool allDone = !fifo_in.notEmpty();
    for(Integer i = 0; i < 2; i = i + 1) allDone = allDone &&& regs[i][0] matches
→ tagged Invalid ? True : False;
    return allDone;
endfunction

for(Integer i = 0; i < 5; i = i + 1) begin
    rule push;
        if (i == 0)
            stages[i].put(fifo_in.first);
        else if (regs[i-1][0] matches tagged Valid .x)
            stages[i].put(x);
        endrule
    end

    rule pull_0 if(regs[0][1] matches tagged Invalid);
        fifo_in.deq();
        let t <- stages[0].result();
        regs[0][1] <= tagged Valid t;
    endrule

    for(Integer i = 1; i < 4; i = i + 1) begin
        rule pull if (regs[i][1] matches tagged Invalid &&& isValid(regs[i-1][0]));
            regs[i-1][0] <= tagged Invalid;
            let t <- stages[i].result();
            regs[i][1] <= tagged Valid t;
        endrule
    end

    rule pull_4 if(isValid(regs[3][0]));
        regs[3][0] <= tagged Invalid;
        let t <- stages[4].result();
        fifo_out.enq(t);
    endrule

    method Action setParam(UInt#(2) addr, Int#(32) val) if(setParamAllowed);
        changeables[addr].setParameter(val);
    endmethod

    interface CalcUnit calc;
        method Action put(Int#(32) x);
            fifo_in.enq(x);
        endmethod

        method ActionValue#(Int#(32)) result();
            fifo_out.deq();
            return fifo_out.first;
        endmethod
    endinterface
endmodule

```

2.3c) Pipelinetypen

Welchen Typ hat die soeben implementierte, modulare Pipeline (dynamisch vs statisch; elastisch vs starr)?

Können Sie in der modularen Pipeline bspw. das mkMult-Teilmodul durch eine alternative Implementierung ersetzen, welche eine variable Anzahl an Takten zur Berechnung des Produkts benötigt (abhängig von den Eingabedaten)¹⁰? Ändert dies etwas am Typ der Pipeline?

Ist dies auch bei der einfachen Pipeline aus Aufgabe 2.1 möglich?

Lösungsvorschlag:

Die modulare Pipeline ist **statisch** und **elastisch**.

Die modulare Pipeline unterstützt Pipelinestufen mit variabler Latenz (wenn eine entsprechende CalcUnit Implementierung genutzt wird). In diesem Fall ist die Pipeline dann **dynamisch** und **elastisch**.

In der einfachen Pipeline aus Aufgabe 2.1 sind Pipelinestufen mit variabler Latenz **nicht** möglich.

Hinweis: Dies gilt so für die vorgestellte Musterlösung. Ihre eigene Implementierung kann eine andere Charakterisierung haben.

2.3d) Zusatzaufgabe

Erstellen Sie eine alternative Implementierung der modularen Pipeline, welche FIFOs zum Speichern der Zwischenergebnisse nutzt (anstatt Registern). Ändert dies etwas am Pipelinetyp? Können Sie auch mit dieser Implementierung CalcUnits mit variabler Latenz nutzen?

Lösungsvorschlag:

Auch die folgende Pipeline ist **statisch** und **elastisch**. Sie unterstützt ebenfalls Pipelinestufen mit variabler Latenz und ist dann **dynamisch** und **elastisch**.

```
module mkPipeline(Pipeline);
  CalcUnitChangeable stage1 <- mkAddA;
  CalcUnitChangeable stage2 <- mkMul;
  CalcUnitChangeable stage3 <- mkMul;
  CalcUnit stage4 <- mkDiv4;
  CalcUnit stage5 <- mkAdd128;

  Vector#(3, CalcUnitChangeable) changeables;
  changeables[0] = stage1;
  changeables[1] = stage2;
  changeables[2] = stage3;

  Vector#(5, CalcUnit) stages;
  stages[0] = stage1.calc;
  stages[1] = stage2.calc;
  stages[2] = stage3.calc;
  stages[3] = stage4;
  stages[4] = stage5;

  Vector#(6, FIFO#(Int#(32))) fifos <- replicateM(mkSizedFIFO(2));
  function Bool fempty(FIFO#(_) f) = !f.notEmpty;
  function setParamAllowed();
    Bool allDone = True;
    for(Integer i = 0; i < 3; i = i + 1) allDone = allDone && fempty(fifos[i]);
    return allDone;
endfunction
```

¹⁰vgl. Foliensatz BSV; Folie 21

```

for(Integer i = 0; i < 5; i = i + 1) begin
  rule push;
  stages[i].put(fifos[i].first);
endrule
end

for(Integer i = 1; i < 6; i = i + 1) begin
  rule pull;
  fifos[i-1].deq;
  let t <- stages[i-1].result();
  fifos[i].enq(t);
endrule
end

method Action setParam(UInt#(2) addr, Int#(32) val) if(setParamAllowed());
  changeables[addr].setParameter(val);
endmethod

interface CalcUnit calc;
  method Action put(Int#(32) x);
  fifos[0].enq(x);
endmethod

  method ActionValue#(Int#(32)) result();
  let res = fifos[5].first();
  fifos[5].deq;
  return res;
endmethod
endinterface
endmodule

```

Aufgabe 2.4: Tagged Unions

Aktuell kann die Pipeline aus Aufgabe 2.3 nur vorzeichenbehaftete Integerwerte verarbeiten. Wir wollen die Module und Interfaces nun so erweitern, dass sowohl vorzeichenbehaftete (signed), als auch nicht-vorzeichenbehaftete (unsigned) Integerwerte verarbeitet werden können. Dabei soll die Vorzeichenbehaftung der Ein- und Ausgabedaten zur Laufzeit überprüfbar sein. Bluespec stellt hierfür als Mittel der Wahl die so genannten **Tagged Unions** zur Verfügung. Diese sind Strukturen, deren Daten zur Laufzeit mit einem **Tag** versehen sind, wodurch die Schaltung interpretieren kann welcher Datentyp gerade anliegt.

Für diese Aufgabe definieren wir das Tagged Union **SignedOrUnsigned** mit

```

typedef union tagged {UInt#(32) Unsigned;
                    Int#(32) Signed;
                    } SignedOrUnsigned deriving(Eq, Bits);

```

Der obige Code erledigt dabei folgende Dinge:

- Er legt ein Tagged Union mit den beiden Tags **Unsigned** und **Signed** an.
- Er setzt den Typnamen dieses Tagged Unions mittels **typedef** auf **SignedOrUnsigned**.
- Er übergibt dem Bluespec Compiler mit **deriving** die Information, dass der Typ mittels **==** vergleichbar ist
- Er übergibt dem BSC die Information, dass der Typ mit Bits darstellbar ist, also die Funktionen **pack** und **unpack** unterstützt, mit.

Wir können diesem Typen nun mit folgendem Code **Signed** oder **Unsigned** Werte zuweisen:

Durch das verwendete **deriving**-Statement erstellt der Compiler automatisch die in Abbildung 2 gezeigte Bitrepräsentation für die beiden Variablen.




Abbildung 2: Standard Bitrepräsentation von Tagged Unions

Der Inhalt des Typen **SignedOrUnsigned** kann mittels Pattern Matching¹¹ interpretiert werden. Nehmen wir an, es gäbe drei Register **a**, **b** und **c**, die jeweils den Typen **SignedOrUnsigned** beinhalten und **a** und **b** seien auf Operanden gesetzt, während **c** das Ergebnis beinhalten soll. Dann könnte man folgende Variante des Pattern Matchings wählen:

```

rule doAdd (pack(a)[32] == pack(b)[32]);
  case (a) matches
    tagged Unsigned .usa: begin
      UInt#(32) usb = unpack(pack(b)[31:0]);
      c <= tagged Unsigned (usa+usb);
    end
    tagged Signed .sa: begin
      Int#(32) sb = unpack(pack(b)[31:0]);
      c <= tagged Signed (sa + sb);
    end
  endcase
  $display("%h", c);
endrule

```

Dabei erledigt der obige Code folgendes:

- Die explizite Guard der Regel überprüft, ob die Tag-Bits der beiden Operandenregister übereinstimmen. Nur in diesem Fall feuert die Rule.
- Das Case-Statement überprüft das Tag-Bit von **a**. Ist das Tag **Unsigned**, wird der **UInt**-Wert in einer Variable namens **usa** gespeichert, andernfalls in **sa**.
- Weiterhin wird der Inhalt des Registers **b** passend zum Tag von **a** in einem **UInt** oder **Int** gespeichert. Dabei wird **b** erst mit **pack** zum Typ **Bit** konvertiert, wodurch wir auf einzelne Bitvektoren zugreifen können. Anschließend wird der 32-Bit Vektor mittels **unpack** zum Typen **UInt** oder **Int** konvertiert und in einer passenden Variable gespeichert.
- Zuletzt wird im Register **c** das Ergebnis der Addition der beiden Registerinhalte und das passende Tag gespeichert.

Eine Alternative hierzu, die uns die Verwendung des expliziten Casts von **b** erspart, wäre die Verwendung von zwei Regeln, die in Ihren Guards direkt das Pattern-Matching mit der Wertzuweisung durchführen.

```
rule doAddSigned (a matches tagged Signed .sa &&& b matches tagged Signed .sb);
    c <= tagged Signed (sa+sb);
endrule
```

```
rule doAddUnsigned (a matches tagged Unsigned .usa &&& b matches tagged Unsigned
  ↪ .usb);
    c <= tagged Unsigned (usa+usb);
endrule
```

¹¹BSV by Example; Kapitel 10.1.6; S. 141f.

2.4a) Interfaces

Erstellen Sie ein Interface **TU_Pipeline** mit dem auf **SignedOrUnsigned** angepassten Interface aus. Legen Sie weiterhin ein Package **CalcUnits** an und definieren Sie dort die Interfaces **CalcUnit** und **CalcUnitChangeable** und passen Sie diese auf den Typen **SignedOrUnsigned** an. Denken Sie dran, dass Sie das Package **CalcUnits** in **TU_Pipeline.bsv** importieren müssen, um die dort deklarierten Interfaces nutzen zu können.

Lösungsvorschlag:

```
// In Datei TU_Pipeline.bsv
interface TU_Pipeline;
    interface CalcUnit calc;
    method Action setParam(UInt#(2) addr, SignedOrUnsigned val);
endinterface

// In Datei CalcUnits.bsv
interface CalcUnit;
    method Action put(SignedOrUnsigned x);
    method ActionValue#(SignedOrUnsigned) result();
endinterface

interface CalcUnitChangeable;
    interface CalcUnit calc;
    method Action setParameter(SignedOrUnsigned param);
endinterface
```

2.4b) Module

Implementieren Sie nun erneut die Pipeline aus Aufgabe 2.3 und Ihre Teilmodule mit den entsprechenden neuen Interfaces. Implementieren Sie dabei die Untermodule im Package **CalcUnits**.

Lösungsvorschlag:

```
// Teilmodule
module mkAddA(CalcUnitChangeable);
    Reg#(SignedOrUnsigned) a <- mkRegU;
    Wire#(SignedOrUnsigned) in <- mkWire;
    Wire#(SignedOrUnsigned) out <- mkWire;

    /* RULES */
    rule computeSigned(a matches tagged Signed .sa &&&
                      in matches tagged Signed .sin);
        out <= tagged Signed (sa + sin);
    endrule

    rule computeUnsigned(a matches tagged Unsigned .usa &&&
                        in matches tagged Unsigned .usin);
        out <= tagged Unsigned (usa + usin);
    endrule

    rule crash(pack(in)[32] != pack(a)[32]); // if we input an unfitting value we crash
        $display("ERROR: mkAddA got wrong input!");
        $display("Tag a: %b, Tag in: %b", pack(a)[32], pack(in)[32]);
        $finish();
endmodule
```

```

endrule

/* INTERFACE DEFINITION */
method Action setParameter(SignedOrUnsigned param);
  a <= param;
endmethod

interface CalcUnit calc;
  method Action put(SignedOrUnsigned x);
    in <= x;
  endmethod

  method ActionValue#(SignedOrUnsigned) result();
    noAction;
    return out;
  endmethod
endinterface
endmodule

module mkMul(CalcUnitChangeable);
  Reg#(SignedOrUnsigned) b <- mkRegU; // or c
  Wire#(SignedOrUnsigned) in <- mkWire;
  Wire#(SignedOrUnsigned) out <- mkWire;

  /* RULES */
  rule computeSigned(b matches tagged Signed .sb &&&
                    in matches tagged Signed .sin);
    out <= tagged Signed (sb * sin);
  endrule

  rule computeUnsigned(b matches tagged Unsigned .usb &&&
                      in matches tagged Unsigned .usin);
    out <= tagged Unsigned (usb * usin);
  endrule

  rule crash(pack(in)[32] != pack(b)[32]);
    $display("ERROR: mkMulB got wrong input!");
    $display("Tag b: %b, Tag in: %b", pack(b)[32], pack(in)[32]);
    $finish();
  endrule

/* INTERFACE DEFINITION */
method Action setParameter(SignedOrUnsigned param);
  b <= param;
endmethod

interface CalcUnit calc;
  method Action put(SignedOrUnsigned x);
    in <= x;
  endmethod

  method ActionValue#(SignedOrUnsigned) result();
    noAction;
    return out;
  endmethod

```

```

    endinterface
endmodule

module mkDiv4(CalcUnit);
    Wire#(SignedOrUnsigned) in <- mkWire;

    method Action put(SignedOrUnsigned x);
        in <= x;
    endmethod

    method ActionValue#(SignedOrUnsigned) result();
        noAction;
        case (in) matches
            tagged Unsigned .usin : return tagged Unsigned (usin / 4);
            tagged Signed .sin : return tagged Signed (sin / 4);
        endcase
    endmethod
endmodule : mkDiv4

module mkAdd128(CalcUnit);
    Wire#(SignedOrUnsigned) in <- mkWire;

    method Action put(SignedOrUnsigned x);
        in <= x;
    endmethod

    method ActionValue#(SignedOrUnsigned) result();
        noAction;
        case (in) matches
            tagged Unsigned .usin : return tagged Unsigned (usin + 128);
            tagged Signed .sin : return tagged Signed (sin + 128);
        endcase
    endmethod
endmodule : mkAdd128

// Pipeline Modul

module mkTU_Pipeline(TU_Pipeline);
    CalcUnitChangeable stage1 <- mkAddA;
    CalcUnitChangeable stage2 <- mkMul;
    CalcUnitChangeable stage3 <- mkMul;
    CalcUnit stage4 <- mkDiv4;
    CalcUnit stage5 <- mkAdd128;

    Vector#(3, CalcUnitChangeable) changeables;
    changeables[0] = stage1;
    changeables[1] = stage2;
    changeables[2] = stage3;

    Vector#(5, CalcUnit) stages;
    stages[0] = stage1.calc;
    stages[1] = stage2.calc;
    stages[2] = stage3.calc;

```



```

stages[3] = stage4;
stages[4] = stage5;

Vector#(6, FIFO#(SignedOrUnsigned)) fifos <- replicateM(mkSizedFIFO(2));

for(Integer i = 0; i < 5; i = i + 1) begin
    rule push;
        stages[i].put(fifos[i].first);
        fifos[i].deq;
    endrule
end

for(Integer i = 1; i < 6; i = i + 1) begin
    rule pull;
        let t <- stages[i-1].result();
        fifos[i].enq(t);
    endrule
end

method Action setParam(UInt#(2) addr, SignedOrUnsigned val);
    changeables[addr].setParameter(val);
endmethod

interface CalcUnit calc;
    method Action put(SignedOrUnsigned x);
        fifos[0].enq(x);
    endmethod

    method ActionValue#(SignedOrUnsigned) result();
        let res = fifos[5].first();
        fifos[5].deq;
        return res;
    endmethod
endinterface
endmodule

```

2.4c) Tests

Wir stellen Ihnen ein Template **U2_4c_CustomTest.bsv** und eine angepasste **U2_4c_Testbench.bsv** zur Verfügung. Das Modul **mkCustomTest** wurde mit Parametern so erweitert, dass man Testfälle einfach bei der Erstellung des Moduls übergeben kann. Innerhalb von **mkCustomTest** können diese Parameter wie gewöhnliche BSV-Variablen genutzt werden.

Erweitern Sie in **mkCustomTest** das Statement **s**, sodass dieses den Parameter **in** in die Pipeline eingibt und danach das Ergebnis der Pipeline ausliest.

Wenn das Ergebnis der Pipeline nicht mit dem Parameter **expOut** übereinstimmt, soll der Text „Test INDEX failed. Expected expOut, got result“ ausgegeben werden. Dabei sollen die beiden Ergebniswerte in hexadezimaler Darstellung angezeigt werden. Der **INDEX** soll dabei der Index des Testfalls sein. Dieser ist der Inhalt des Parameters **in**.

Wenn das erwartete Ergebnis mit dem tatsächlichen Ergebnis übereinstimmt, soll der Text „Test INDEX succeeded!“ ausgegeben werden. Der **INDEX** entspricht auch hier dem Index des Testfalls.

Lösungsvorschlag:

```
Stmt s = {
  seq
    setup(dut, a, b, c);
    action
      dut.calc.put(in);
    endaction
    action
      let res <- dut.calc.result();
      if (res != expOut) begin
        $display("Test %d failed. Expected %h, got %h", pack(in)[31:0], expOut,
→      res);
      end
      else $display("Test %d succeeded!", pack(in)[31:0]);
    endaction
  endseq
};
```

Aufgabe 2.5: Zusatzaufgabe: ALU mit FSM-Testbench

Das Modul mkAutoFSM eignet sich hervorragend zur Erstellung von Testbenches.

Schreiben Sie eine Testbench für das Modul mkSimpleALU aus der ersten Übung. Nutzen Sie dabei einen Vektor, der alle Testdaten beinhaltet. Lagern Sie häufig genutzte Teile (Operanden eingeben und Ergebnis überprüfen) in eine extra FSM aus, indem Sie mkAutoFSM und mkFSM kombinieren. Einen Vektor können Sie folgendermaßen erzeugen:

```
1  typedef struct {
2      Int#(32) opA;
3      Int#(32) opB;
4      AluOps   operator;
5      Int#(32) expectedResult;
6  } TestData deriving (Eq, Bits);
7  ...
8  Vector#(20, TestData) myVector;
9  myVector[0] = TestData {opA: 2, opB: 4, operator: Add, expectedResult: 6};
10 ...
11 myVector[19] = TestData {opA: 4, opB: 2, operator: Div, expectedResult: 2};
```

Lösungsvorschlag:

```
import Vector::*;

typedef enum {Mul, Div, Add, Sub, And, Or, Pow} AluOps deriving (Eq, Bits, FShow);

typedef struct {
  Int#(32) opA;
  Int#(32) opB;
  AluOps   operator;
  Int#(32) expectedResult;
} TestData deriving (Eq, Bits);
```

```

module mkAluFSMTB(Empty);

Vector#{5, TestData) myVector;
myVector[0] = TestData {opA: 2, opB: 4, operator: Add, expectedResult: 6};
myVector[1] = TestData {opA: 2, opB: 4, operator: Mul, expectedResult: 8};
myVector[2] = TestData {opA: 4, opB: 2, operator: Div, expectedResult: 2};
myVector[3] = TestData {opA: 4, opB: 0, operator: Pow, expectedResult: 1};
myVector[4] = TestData {opA: 4, opB: 4, operator: Pow, expectedResult: 256};
Reg#{UInt#{32}) dataPtr <- mkReg(0);

HelloALU uut <- mkHelloALU();
Stmt checkStmt = {
  seq
    action
      let currentData = myVector[dataPtr];
      uut.setupCalculation(currentData.operator, currentData.opA, currentData.opB);
    endaction
    action
      let currentData = myVector[dataPtr];
      let result <- uut.getResult();
      let print = $format("Calculation: %d ", currentData.opA) +
        fshow(currentData.operator) + $format("%d", currentData.opB);
      $display(print);
      if(result == currentData.expectedResult) begin
        $display("Result correct: %d", result);
      end else begin
        $display("Result incorrect: %d != ", result, currentData.expectedResult);
      end
    endaction
  endseq
};
FSM checkFSM <- mkFSM(checkStmt);
Stmt mainFSM = {
  seq
    for(dataPtr <= 0; dataPtr < 5; dataPtr <= dataPtr + 1) seq
      checkFSM.start();
      checkFSM.waitTillDone();
    endseq
  endseq
};
mkAutoFSM(mainFSM);
endmodule

```

Aufgabe 2.6: Zusatzaufgabe: ALU mit Tagged Unions

Erweitern Sie die ALU aus Übung 1, sodass sie Signed und Unsigned Integer verarbeiten kann.

Lösungsvorschlag:

```

typedef enum {Mul, Div, Add, Sub, And, Or, Pow} AluOps deriving (Eq, Bits);
typedef union tagged {UInt#{32} Unsigned; Int#{32} Signed;} SignedOrUnsigned
  ↳ deriving(Bits, Eq);

```

```

interface Power;
  method Action setOperands(SignedOrUnsigned a, SignedOrUnsigned b);
  method Int#(32) getResultSigned();
  method UInt#(32) getResultUnsigned();
endinterface

module mkPower(Power);
  Reg#(Bool) resultValid <- mkReg(False);

  Reg#(SignedOrUnsigned) opA <- mkRegU;
  Reg#(SignedOrUnsigned) opB <- mkRegU;
  Reg#(Int#(32)) result_signed <- mkReg(1);
  Reg#(UInt#(32)) result_unsigned <- mkReg(1);

  rule calc (pack(opB)[31:0] > 0);
    Bit#(33) b_opB = pack(opB);
    opB <= unpack({b_opB[32], b_opB[31:0] - 1});
    result_signed <= result_signed * unpack(pack(opA)[31:0]);
    result_unsigned <= result_unsigned * unpack(pack(opA)[31:0]);
  endrule

  rule calcDone (pack(opB)[31:0] == 0 && !resultValid);
    resultValid <= True;
  endrule

  method Action setOperands(SignedOrUnsigned a, SignedOrUnsigned b);
    result_signed <= 1;
    result_unsigned <= 1;
    opA <= a;
    opB <= b;
    resultValid <= False;
  endmethod

  method Int#(32) getResultSigned() if(resultValid);
    return result_signed;
  endmethod

  method UInt#(32) getResultUnsigned() if(resultValid);
    return result_unsigned;
  endmethod
endmodule

interface HelloALU;
  method Action setupCalculation(AluOps op, SignedOrUnsigned a, SignedOrUnsigned b);
  method ActionValue#(SignedOrUnsigned) getResult();
endinterface

module mkHelloALU(HelloALU);
  Reg#(Bool) newOperands <- mkReg(False);
  Reg#(Bool) resultValid <- mkReg(False);
  Reg#(AluOps) operation <- mkReg(Mul);
  Reg#(SignedOrUnsigned) opA <- mkReg(tagged Signed 0);
  Reg#(SignedOrUnsigned) opB <- mkReg(tagged Signed 0);
  Reg#(SignedOrUnsigned) result <- mkReg(tagged Signed 0);

```

```

Power pow <- mkPower();

rule calculateSigned (opA matches tagged Signed .va &&& opB matches tagged Signed
→ .vb &&& newOperands);
    Int#(32) rTmp = 0;
    case(operation)
        Mul: rTmp = va * vb;
        Div: rTmp = va / vb;
        Add: rTmp = va + vb;
        Sub: rTmp = va - vb;
        And: rTmp = va & vb;
        Or: rTmp = va | vb;
        Pow: rTmp = pow.getResultSigned();
    endcase
    result <= tagged Signed rTmp;
    newOperands <= False;
    resultValid <= True;
endrule

rule calculateUnsigned (opA matches tagged Unsigned .va &&& opB matches tagged
→ Unsigned .vb &&& newOperands);
    UInt#(32) rTmp = 0;
    case(operation)
        Mul: rTmp = va * vb;
        Div: rTmp = va / vb;
        Add: rTmp = va + vb;
        Sub: rTmp = va - vb;
        And: rTmp = va & vb;
        Or: rTmp = va | vb;
        Pow: rTmp = pow.getResultUnsigned();
    endcase
    result <= tagged Unsigned rTmp;
    newOperands <= False;
    resultValid <= True;
endrule

function Bool isUnsigned(SignedOrUnsigned v);
    if(v matches tagged Unsigned .va) return True;
    else return False;
endfunction

rule dumpInvalid (newOperands && isUnsigned(opA) != isUnsigned(opB));
    $display("Invalid combination of Signed and Unsigned Operands");
    newOperands <= False;
    resultValid <= False;
endrule

method Action setupCalculation(AluOps op, SignedOrUnsigned a, SignedOrUnsigned b)
→ if(!newOperands);
    opA <= a;
    opB <= b;
    operation <= op;
    newOperands <= True;
    resultValid <= False;
    if(op == Pow) begin

```

```

        pow.setOperands(a, b);
    end
endmethod

method ActionValue#(SignedOrUnsigned) getResult() if(resultValid);
    resultValid <= False;
    return result;
endmethod
endmodule

module mkALUTestbench(Empty);
    HelloALU uut <- mkHelloALU();
    Reg#(UInt#(8)) testState <- mkReg(0);

    rule checkMul (testState == 0);
        uut.setupCalculation(Mul, tagged Unsigned 4, tagged Unsigned 5);
        testState <= testState + 1;
    endrule

    rule checkDiv (testState == 2);
        uut.setupCalculation(Div, tagged Unsigned 12, tagged Unsigned 4);
        testState <= testState + 1;
    endrule

    rule checkAdd (testState == 4);
        uut.setupCalculation(Add, tagged Unsigned 12, tagged Unsigned 4);
        testState <= testState + 1;
    endrule

    rule checkSub (testState == 6);
        uut.setupCalculation(Sub, tagged Unsigned 12, tagged Unsigned 4);
        testState <= testState + 1;
    endrule

    rule checkAnd (testState == 8);
        uut.setupCalculation(And, tagged Unsigned 32'hA, tagged Unsigned 32'hA);
        testState <= testState + 1;
    endrule

    rule checkOr (testState == 10);
        uut.setupCalculation(Or, tagged Unsigned 32'hA, tagged Unsigned 32'hA);
        testState <= testState + 1;
    endrule

    rule checkPow (testState == 12);
        uut.setupCalculation(Pow, tagged Unsigned 2, tagged Unsigned 12);
        testState <= testState + 1;
    endrule

    rule checkMulSigned (testState == 14);
        uut.setupCalculation(Mul, tagged Signed 4, tagged Signed -5);
        testState <= testState + 1;
    endrule

    rule checkDivSigned (testState == 16);

```

```

    uut.setupCalculation(Div, tagged Signed -12, tagged Signed -4);
    testState <= testState + 1;
endrule

rule checkAddSigned (testState == 18);
    uut.setupCalculation(Add, tagged Signed -12, tagged Signed 4);
    testState <= testState + 1;
endrule

rule checkSubSigned (testState == 20);
    uut.setupCalculation(Sub, tagged Signed 12, tagged Signed -4);
    testState <= testState + 1;
endrule

rule checkAndSigned (testState == 22);
    uut.setupCalculation(And, tagged Signed 32'hA, tagged Signed 32'hA);
    testState <= testState + 1;
endrule

rule checkOrSigned (testState == 24);
    uut.setupCalculation(Or, tagged Signed 32'hA, tagged Signed 32'hA);
    testState <= testState + 1;
endrule

rule checkPowSigned (testState == 26);
    uut.setupCalculation(Pow, tagged Signed -2, tagged Signed 11);
    testState <= testState + 1;
endrule

rule printResults (unpack(pack(testState)[0]));
    let result <- uut.getResult();
    if (result matches tagged Signed .r) begin
        $display("Result (Signed): %d", r);
    end else if (result matches tagged Unsigned .r) begin
        $display("Result (Unsigned): %d", r);
    end
    testState <= testState + 1;
endrule

rule endSim (testState == 28);
    $finish();
endrule
endmodule

```

Anm.: Sie müssen den präsentierten Code mittels `-aggressive-conditions` kompilieren, da ansonsten das `mkPower`-Modul implizit blockiert.